
BEDEUTUNG DES GESAMTSCHADENS 

Der Gesamtschaden als Summe aller Schäden für festes Kollektiv innerhalb eines festen Zeitraumes 
Verteilung wichtig für 

⇒ Franchisen 
⇒ Rückversicherung 
⇒ Ruinwahrscheinlichkeiten 
⇒ Reserveberechnungen 

 

Seine Verteilung ist bedeutsam für die Gesamtbeurteilung des Portefeuilles und stellt daher eine wesentliche 
Aufgabe innerhalb der Risikotheorie dar. 
 

KOLLEKTIVES MODELL DER RISIKOTHEORIE  

Dieses Modell läßt sich wie folgt beschreiben : 

Z ufallsvariab le :

N pn n:=  Schadenanzahl im Kollektiv in Periode mit Zähldichte 
0

Es betrachtet nur tatsächliche Schadenereignisse, ohne dabei auf die diesen Schäden zugrunde liegenden 
versicherungstechnischen Einheiten einzugehen. Insbesondere wird nicht betrachtet, welche Risiken keine 
Schäden, welche mehrere und wieviele Schäden verursachen.

EIGENSCHAFTEN 

Die folgenden wichtigen Eigenschaften gelten für den Gesamtschaden und seine Verteilung: 

Mit der Angabe als gewichteter unendlicher Summe von Faltungen sind die Berechungsprobleme 
offenbar noch nicht gelöst. Die Relation zwischen den momenterzeugenden Funktionen kann zur Herleitung 
der Eigenschaften von Erwartungswert und Varianz genutzt werden und findet auch bei einigen 
verteilungsunabhängigen Verfahren zur Berechnung der Gesamtschadenverteilung  Anwendung. 

SY STEMATISIERUNG DER BERECHNUNG 

Die nachfolgende Grafik beschreibt die wesentlichen Verfahrensschritte und klassifiziert die derzeitig 
gebräuchlichsten Verfahren : 

Stichprobe 
aus m Perioden

Auswertung der Stichprobe
Momentenschätzung
Stichprobenkennzahlen

Verteilungsunabhängige Verfahren

Verteilung von N, X sind unbekannt
erste Momente von S werden geschätzt

Orthogonale Polynome
Edgeworth-Approximation
Normal-Power-Approximation
Bootstrap-Methode

Ableiten von Verteilungshypothesen
für N, X ; Anpassungstests

Verteilungsabhängige Verfahren

Verteilung von N, X sind als bekannt
vorausgesetzt (d.h. zuvor angepaßt)
Diskretisierung Einzelschadenhöhe

Simulation
Esscher-Verfahren
TD-Techniken
rekursive Verfahren
Fast-Fourier-Methode  

PANJER-REKURSION 

Wir betrachten im folgenden eine einfache Ausprägung des rekursiven Ansatzes: 

Schlüssel ist die Beobachtung, daß die Zähldichten der gängigen Anzahlverteilungen einer bestimmten 
Rekursion genügen. 

Die Zähldichten der Anzahlverteilungen
|Poisson mit Parameter λ

|Binomial mit Parametern n, p

|Negativ Binomial mit den Parametern r und p

erfüllen eine Rekursion der Form

p a b
n

pn n= + ⋅ −( ) 1   (Panjer)

(es gibt eine Vielzahl von Verallgemeinerungen)

a b= =0  λ

a b np
p

p
p= − = ⋅ +− −1 1 1  ( )

a p b p r= = ⋅ −   ( )1

 

Die Erweiterungen des Ansatzes betreffen die Ordnung der Rekursionsgleichung und verlangen die 
Gültigkeit erst ab einer festen Stelle an. Damit werden unter Erhalt der rekursiven Berechnungsmöglichkeit 
komplexere Verteilungen an die Schadenanzahl anpaßbar. (z. B. Delaport-Verteilung) 

REKURSIONSFORMEL 

Mit Hilfe der Panjer-Rekursion und des Zusammenhangs der momenterzeugenden Funktionen läßt sich 
eine Rekursion für die Dichte des Gesamtschadens herleiten. 

N pn n sei Zufallsvariable mit Zähldichte ( )  ,
die die Panjer - Rekursion erfüllt.

mit dem Rekursionsanfang:  g f0 0=
 

Dies ist die Basis für den nachfolgenden Algorithmus . 

DAS SAS-PROGRAMM (MIT IML)  

Die Umsetzung der dargestellten Rekursion erfolgt innerhalb des SAS-Systems mit der interaktiven 
Matrixsprache, die ein hohes Maß anFlexibilität und Mächtigkeit aufweist.  

Nachfolgend das komplette kommentierte Programm am Beispiel einer künstlich erzeugten 
Schadenhöhenverteilung. 

/* Bereitstellung einer theoretischen Einzelschadenhöhendatei  mit Mittel 100 */ 
 /* im Praxisprozess nach Modellierung der Einzelschadenhöhen */ 
 
data schh ; 
     do j=1 to 1000; 
 
     k=poisson(200,j); 
     i=j-100; 
     w=k-lag(k); 
     output; 
     end; 
     run; 
 
/* zu kleine Wahrscheinlichkeiten entfernt */ 
 
data schh (keep= i w); 
set schh; 
if i>0; 
run; 
 
/* Ende der Bereitstellung einer Einzelschadenhöhendatei */ 
 
proc imL; 
 
/* Einlesen der SAS-Test-Datei in IML*/ 
 
close schh; 
/* vorsichtshalber schließen, um die richtige Zeigerposition zu haben */ 
 
use schh; 
read current var {w} into snull; 
/* Wahrscheinlichkeit für Schäden der Höhe Null gesondert einlesen */ 
 
read after   var {w} into s; 
/* Rest in den Vektor s einlesen */ 
 
/* Ergebnisvektor initialisieren */ 
f=0; 
 
/* erstes und zweites Moment der Schadenhöhenverteilung berechnen */ 
 
m1= (1:nrow(s))* s ; 
m2= ((1:nrow(s))#(1:nrow(s))) *s ; 
 
/* Type der Schadenanzahlverteilung festlegen */ 
/* zur Auswahl stehen poisson und negbin   */ 

typ="negbin"; 
 
/* Paramter für den Typ der Schadenanzahlverteilung setzen */ 
/* Mittelwert und Varianz der Schadenanzahl */ 
 
manz=100; 
vanz=250; 
 
/* Poisson-Fall */ 
if typ="poisson" then do; 
                      a=0; 
                      b=manz; 
                      f[1]=exp(b*snull-b); 
                      /* Ergebnisvektor f initialisieren */ 
                      /* f[1]= Prob [gesaufw =0]         */ 
                      /* entsteht durch Diskretisierung  */ 
                      mges=b*m1; 
                      vges=b*m2; 
 
                      end; 
 
/* negbin - Fall */ 
if typ="negbin" then do; 
                      /* falls die Parameter nach der Momenten-Methode */ 
                      /* berechnet werden sollen, sonst direkt setzen  */ 
                      a=(vanz-manz)/vanz; 
                      b=(manz*manz+manz-vanz) /vanz; 
                      f[1]=(1+ (a*((a+b)/(1-a))*(1-snull))/(a+b))##((-a-b)/a); 
                      /* Ergebnisvektor f initialisieren */ 
                      /* f[1]= Prob [gesaufw =0]         */ 
                      /* entsteht durch Diskretisierung  */ 
                      mges=(a+b)/(1-a)*m1; 
                      vges=((a+b)/(1-a))*m2+m1##2*(a*(a+b))/(1-a)##2; 
 
                      end; 
 
 
 
 
 
/* jetzt läßt sich etwa die 3 sigma -Regel anwenden */ 
 
/* drei-sigma-Regel */ 
do j=1 to ceil(mges+3*sqrt(vges)); 
 
/* Anzahl der Werte hier sinnvoll festzulegen */ 
 
 k=min(j,nrow(s)); 
 
/* stets nur solange wie Einzelschadenhäufigkeiten vorhanden */ 
 
 p=T(1:k) * b/j+a || s[1:k] || f[1:k]; 
/* eine dreispaltige Matrix wird aufgebaut */ 
/* sie enthält die Gewichtungsfaktoren, die Einzelschadenhöhen und */ 
/* die bisher berechneten Gesamtschadenhöhen */ 
 
 x=p[,#][+,]/(1-a*snull); 
/* die Summe über die Produkte wird mit den "subscript reduction operators" gebildet */ 

 f=x//f; 
/* der kleine Trick ! */ 
/* f wird in Kellerstruktur abgelegt */ 
/* es ist so  keine komplexen Indizes zu führen */ 
/* alles steht richtig bereit für den nächsten Schritt ! */ 
 
end; 
 
 
 
 
/* Abspeichern des Ergebnisses in der Datei work.erges */ 
 
create erges from f; 
append from f; 
 
quit ; 
 
/* IML verlassen */ 
 
/* die Reihenfolge ist wegen der Kellerstruktur nun umzukehren */ 
 
/* Index k einführen */ 
data erges; 
set erges; 
k=_N_; 
run; 
 
/* absteigend nach k sortieren */ 
 proc sort data=erges; by descending k;run; 
 
/* Ergebnisdatei indiziert ablegen, überflüssigen Index löschen */ 
data erges (drop=k); 
set erges (rename=(col1=wges)); 
i=_N_ -1; 
 
run; 
 
 
/* indiziert mit Index i hat man nun */ 
/* wges= Prob( Gesamtsschaden= i )   */ 
 
/* grafische Darstellung des Ergebnisses */ 
symbol1 interpol=join value=dot height=0.3; 
proc gplot data=erges; 
 plot  wges*i; 
 run; 
quit; 

ZUSAMMENFASSUNG 

Für Gesamtschadenverteilung haben Sie erhalten : 

9Einführung in Problemstellung 

9eine elegante und schnelle theoretische Lösung 

9praktische Umsetzung in SAS mit gutem Laufzeitverhalten 
 
Bei dem Gesamtprozess werden in den ersten beiden Schritten Anpassungsfehler und Diskretisierungsfehler 
auftreten. Dann ist die Berechnung exakt. 
Bei mehreren  Teilportefeuillles sind diese durch Faltung zusammenzufassen. 
Hier haben wir gute Erfahrungen bei der Nutzung der schnellen Fouriertransformationen in SAS gemacht. 

